Benthic Larval In-Situ Sampler (BLISS) Pump Testing

P. McGill

9Oct19 - Tests with pump but no filters

13Aug20 - Tests with pump and filters in final configuration

Tests with pump but no filters

Enter the data from tank testing of the pump. Each measurement is {RPM, flow, power}.

In[1]:=
```
data = {{0, 0, 1.9}, {100, 20, 3.0}, {200, 48, 5.4}, {300, 74, 10.9}, {400, 97, 21},
   {500, 127, 38}, {600, 160, 64}, {700, 188, 103}, {800, 216, 160}};
```

In[2]:=
```
MatrixForm@data
```

Out[2]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 1.9 \\ 100 & 20 & 3. \\ 200 & 48 & 5.4 \\ 300 & 74 & 10.9 \\ 400 & 97 & 21 \\ 500 & 127 & 38 \\ 600 & 160 & 64 \\ 700 & 188 & 103 \\ 800 & 216 & 160 \end{pmatrix}$$

Note that the pump won’t exceed 800 RPM with a 24 V power supply.

Extract just the pump power as a function of flow and plot the result.

In[3]:=
```
pwrAtFlow = data[[All, {2, 3}]]
```

Out[3]=
```
{{0, 1.9}, {20, 3.}, {48, 5.4}, {74, 10.9},
 {97, 21}, {127, 38}, {160, 64}, {188, 103}, {216, 160}}
```

In[4]:=
```
plot = ListPlot[pwrAtFlow, PlotStyle -> Red,
  GridLines → Automatic, Frame → True, FrameLabel →
   {{"Power (W)", None}, {"Flow (l/min)", "BLISS Larval Pump"}}, ImageSize → Large]
```

Out[4]=



Fit a cubic to the data and plot it with the data.

In[5]:=
```
cubic = Fit[pwrAtFlow, {1, flow, flow^2, flow^3}, flow]
```

Out[5]= $0.910603 + 0.158257\,\text{flow} - 0.00160666\,\text{flow}^2 + 0.0000197041\,\text{flow}^3$

In[6]:=
```
Show[plot, Plot[cubic, {flow, 0, 216}]]
```

Out[6]=



Find the total energy in the 24 V, 40 A-h battery. Assume 80% depth of discharge to derate for low temperature and so the battery will last more than a couple of deployments.

In[7]:=
```
battEnergy = 24 × 40 × 0.8
```

Out[7]= 768.

Define a function to calculate the total volume pumped as a function of the battery energy and flow rate.

In[8]:= totalVol[energy_, flow_] := $\frac{\text{energy}}{\text{cubic}}$ flow 60

Plot the total volume pumped as a function of the chosen flow rate.

In[9]:=
```
Plot[totalVol[battEnergy, flow], {flow, 0, 216}, GridLines → Automatic,
  Frame → True, FrameLabel → {{"Total Volume (l)", None}, {"Flow (l/min)",
     "BLISS Larval Pump Total Volume\rfor 960 W-hr Battery"}}, ImageSize → Large]
```

Out[9]=



Find the flow rate at the maximum total flow.

In[10]:=
```
FindMaximum[totalVol[battEnergy, flow], {flow, 0, 216}]
```

Out[10]=
```
{316925., {flow → 50.0089}}
```

Find how many hours it will take to pump the maximum volume.

In[11]:=
$$\frac{\texttt{battEnergy}}{\texttt{cubic /. flow -> 50}}$$

Out[11]=
```
105.642
```

In[12]:=
```
% / 24
```

Out[12]=
```
4.40174
```

Account for the power consumed by the Raspberry Pi and other loads, i.e. the hotel load. With the Raspberry Pi, relay board, and motor controller, this was measured to be 6.5 W.

In[13]:=
```
hotelPwr = 6.5
```

Out[13]=
```
6.5
```

In[14]:=
$$\texttt{totalVol2[energy\_, hotel\_, flow\_] :=} \frac{\texttt{energy}}{\texttt{cubic + hotel}} \texttt{60 flow}$$

In[15]:=
```
Plot[totalVol2[battEnergy, hotelPwr, flow], {flow, 0, 216},
 GridLines → Automatic, Frame → True, FrameLabel → {{"Total Volume (l)", None},
   {"Flow (l/min)", "BLISS Larval Pump Total Volume\rfor 960
      W-hr Battery with 1.5 W Hotel Load"}}, ImageSize → Large]
```

Out[15]=



Find the flow rate at the maximum total volume.

In[16]:=
```
FindMaximum[totalVol2[battEnergy, hotelPwr, flow], {flow, 0, 216}]
```

Out[16]=
```
{186261., {flow → 74.5788}}
```

Find how many hours it will take to pump the maximum volume, accounting for the hotel load .

In[17]:=
$$\frac{\text{battEnergy}}{\text{hotelPwr} + \text{cubic} \;/.\; \text{flow} \to 58.6}$$

Out[17]=
```
50.7522
```

Plot a dynamic graph that allows you to vary the hotel load to see how it affects the total volume pumped. This graph isn’t dynamic in the PDF version of this document.

In[45]:=
```
Manipulate[Plot[totalVol2[battEnergy, hotel, flow], {flow, 0, 216},
  GridLines → Automatic, PlotRange → {{0, 216}, {0, 400 000}}, Frame → True,
  FrameLabel → {{"Total Volume (l)", None}, {"Flow (l/min)",
     "BLISS Larval Pump Total Volume\rfor 960 W-hr Battery at 80% DOD with
       Variable Hotel Load"}}, ImageSize → Large], {{hotel, 6.5}, 0, 10}]
```

Out[45]=



Due to cruise logistics, we may have only two days per deployment to run the BLISS system. Find the total volume pumped in two days, accounting for the hotel load. First find how much power is available for pumping.

In[19]:= $\text{pumpPwr} = \frac{\text{battEnergy}}{48} - \text{hotelPwr}$

Out[19]= 9.5

Find the flow rate at the given power.

In[20]:=
```
flow48 = Solve[cubic == 13.5, flow]
```

Out[20]= {{flow → 0.550752 − 89.1223 i}, {flow → 0.550752 + 89.1223 i}, {flow → 80.4375}}

Note that only the third solution is real, which is the one we want to use.

Find the total volume pumped in two days.

In[21]:=
```
48 × 60 flow48[[3, 1, 2]]
```
Out[21]=
```
231 660.
```

Key points:

• We can pump hundreds of thousands of liters with one battery in a few days

• The optimum pumping speed depends on the total load on the battery

• We'll have to measure the flow with the carousel and filters in place as they'll surely affect the flow

• The restriction caused by the filters will reduce the time needed to pump the maximum volume

Tests with pump and filters in final configuration

Enter the data from tank testing of the pump and filters. Both 63 and 100 μm filters were tested. Each measurement is {speed (RPM), flow (L/min), power (W)}. Hotel load is the power consumed with the Elmo pump motor drive enabled but with the motor turning at 0 RPM, and includes the Raspberry Pi and relay board; all the other power values include the hotel load plus the power needed to spin the impeller at the given RPM.

In[22]:=
```
data63 = {{0, 0, 5.6}, {150, 17, 8.68}, {200, 27, 11.2},
   {300, 45, 20.44}, {400, 60, 39.2}, {450, 68, 55.44}};
```

In[23]:=
```
MatrixForm@data63
```
Out[23]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 5.6 \\ 150 & 17 & 8.68 \\ 200 & 27 & 11.2 \\ 300 & 45 & 20.44 \\ 400 & 60 & 39.2 \\ 450 & 68 & 55.44 \end{pmatrix}$$

In[24]:=
```
data100 = {{0, 0, 5.6}, {150, 19, 8.68}, {200, 27, 11.2},
   {300, 45, 20.44}, {400, 64, 38.92}, {450, 75, 55.44}};
```

In[25]:=
```
MatrixForm@data100
```
Out[25]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 5.6 \\ 150 & 19 & 8.68 \\ 200 & 27 & 11.2 \\ 300 & 45 & 20.44 \\ 400 & 64 & 38.92 \\ 450 & 75 & 55.44 \end{pmatrix}$$

Extract just the system power as a function of flow and plot the result.

In[26]:=
```
pwrAtFlow63 = data63[[All, {2, 3}]]
```
Out[26]=
```
{{0, 5.6}, {17, 8.68}, {27, 11.2}, {45, 20.44}, {60, 39.2}, {68, 55.44}}
```

In[27]:=
```
pwrAtFlow100 = data100[[All, {2, 3}]]
```
Out[27]=
```
{{0, 5.6}, {19, 8.68}, {27, 11.2}, {45, 20.44}, {64, 38.92}, {75, 55.44}}
```

In[28]:=
```
plot = ListPlot[{pwrAtFlow63, pwrAtFlow100},
  PlotStyle → {Red, Blue}, GridLines → Automatic, Frame → True,
  FrameLabel → {{"Power (W)", None}, {"Flow (L/min)", "BLISS Larval Pump"}},
  PlotLegends → {"63 μm", "100 μm"}, ImageSize → Large]
```

Out[28]=



Fit cubics to the data and plot them with the data.

In[29]:= cubic63 = Fit[pwrAtFlow63, $\{1, flow, flow^2, flow^3\}$, flow]

Out[29]= $5.5537 + 0.3073\,flow - 0.0106892\,flow^2 + 0.00024915\,flow^3$

In[30]:= cubic100 = Fit[pwrAtFlow100, $\{1, flow, flow^2, flow^3\}$, flow]

Out[30]= $5.57757 + 0.129391\,flow + 0.000384821\,flow^2 + 0.0000899544\,flow^3$

In[31]:=
```
Show[plot, Plot[cubic63, {flow, 0, 68}, PlotStyle → {Thick, LightRed}],
  Plot[cubic100, {flow, 0, 75}, PlotStyle → {Thick, LightBlue}], ImageSize → Large]
```

Out[31]=



Define functions to calculate the total volume pumped as a function of the battery energy and flow rate.

In[32]:=
```
totalVol63[energy_, flow_] := energy/cubic63 flow 60;
totalVol100[energy_, flow_] := energy/cubic100 flow 60;
```

In[34]:=
```
Plot[{totalVol63[battEnergy, flow], totalVol100[battEnergy, flow]},
 {flow, 0, 75}, GridLines → Automatic, PlotStyle → {Red, Blue}, Frame → True,
 FrameLabel → {{"Total Volume (L)", None}, {"Flow Rate (L/min)",
    "BLISS Larval Pump Total Volume\rfor 960 W-hr Battery at 80% DOD"}},
 PlotLegends → {"63 μm", "100 μm"}, ImageSize → Large]
```

Out[34]=



Find the flow rates at the maximum total flow.

In[35]:=
```
max63 = FindMaximum[totalVol63[battEnergy, flow], {flow, 0, 75}]
```

Out[35]= {116 981., {flow → 32.2004}}

In[36]:=
```
max100 = FindMaximum[totalVol100[battEnergy, flow], {flow, 0, 75}]
```

Out[36]= {113 034., {flow → 30.7175}}

The two optimal flow rates are so close that we'll just run both filters at the optimal rate for the 63 μm filter.

Find how many hours it will take to pump the maximum volume .

In[37]:=
$$\frac{\texttt{battEnergy}}{\texttt{cubic63 /. flow → max63[[2, 1, 2]]}}$$

Out[37]= 60.5483

Due to cruise logistics, we have only 24 hours to run the BLISS system. Find the total volume pumped in one day, accounting for this constraint. First find how many watts will drain the battery in 24 hours.

In[38]:=
```
pwr24 = battEnergy/24
```
Out[38]= 32.

Find the flow rate at the given power.

In[39]:=
```
flow63 = Solve[cubic63 == pwr24, flow][[3, 1, 2]]
```
Out[39]= 55.3048

Find the total volume pumped in one day.

In[40]:=
```
24 × 60 flow63
```
Out[40]= 79 639.

Find the RPM setting that will achieve this flow rate.

In[41]:=
```
flowAtRPM63 = data63[[All, {1, 2}]]
```
Out[41]= {{0, 0}, {150, 17}, {200, 27}, {300, 45}, {400, 60}, {450, 68}}

In[42]:=
```
cubicRPMvsFlow63 = Fit[flowAtRPM63, {1, rpm, rpm^2, rpm^3}, rpm]
```
Out[42]= $-0.103121 + 0.0619912\,\text{rpm} + 0.000475035\,\text{rpm}^2 - 6.20212 \times 10^{-7}\,\text{rpm}^3$

In[43]:=
```
Show[ListPlot[flowAtRPM63, Frame → True, GridLines → Automatic,
  FrameLabel → {{"Flow Rate (L/min)", None}, {"Impeller Speed (RPM)",
     "BLISS Larval Pump Flow Rate\ras a function of Impeller Speed"}}],
 Plot[cubicRPMvsFlow63, {rpm, 0, 450}, PlotStyle → {Thick, LightRed}],
 ImageSize → Large]
```

Out[43]=



In[44]:=
```
rpm63 = Solve[cubicRPMvsFlow63 == flow63 && 0 < rpm < 450, rpm]
```

Out[44]=
```
{{rpm → 362.854}}
```